
$$ 
$$ Programmierer:       Walter Hogger 
$$  
$$ Erstelldatum:        20.10.1998 
$$ 
 
 
$$--- Ueberpruefung Aktives Part 
 IFTHEN/&ACTPRT == 1 
   MESSG/'Das Programm benoetigt 
ein aktives Part.', $ 
 'Bitte erst ein Part oeffnen !' 
   JUMP/ende: 
 ENDIF 
 
 IFTHEN/&WBORD == &YES 
  &WBORD = &NO 
 ELSE 
  &WBORD = &YES 
 ENDIF 

UNIGRAPHICS 
 

GRIP - Handbuch 
 

[HBB_GRIP_Handbuch.doc]   
 

 
 
GRIP steht für GRaphics Interactive Programming. 
 
Die Sprache wurde generiert, um Ingenieuren und Konstrukteuren die Möglichkeit zu bieten, 
spezielle Abläufe, Funktionen und Vorgehensweisen in Programmform zu bringen. Die Spra-
che wurde nicht für professionelle Programmierer entwickelt, ist also eher „primitiv“ gehalten 
(aus der Sicht von professionellen Programmierern). 
 
Wer GRIP-Programme entwickeln will, muß UNIGRAPHICS kennen! GRIP bietet sehr viele 
Sprachworte, welche Aktionen auslösen oder Objekte erzeugen, die der „UNIGRAPHICS-
Normalbenutzer“ auch interaktiv erzeugen könnte. Wer nicht weiß, was UNIGRAPHICS für 
Möglichkeiten bietet, tut sich auch schwer, in GRIP die entsprechenden Programmwünsche 
umzusetzen. Daraus leitet sich die Empfehlung ab, daß vor der Erstellung eines Programm-
teils erst in UNIGRAPHICS  die Funktion interaktive durchforstet werden sollte, da sehr 
viele – nicht alle! - GRIP-Befehle genauso aufgebaut sind wie die interaktiven Befehle (glei-
che Optionen, gleicher Variationen, etc.). 
 
Nicht für alle interaktiven Funktionen von UNIGRAPHICS gibt es entsprechende GRIP-
Kommandos. Umgekehrt kann man in GRIP einige Aktionen durchführen, an die der normale 
UNIGRAPHICS-Benutzer nicht herankommt. 
 
GRIP-Programme müssen kompiliert und gelinkt (gebunden) werden, um für den Anwender 
aufrufbar zu werden.  
 

HBB Engineering GmbH
Salzstraße 9

D-83454 Anger

Telefon  08656-98488-0
Telefax  08656-98488-88

Info@HBB-Engineering.de
www.HBB-Engineering.de

Zum Ablauf der Programme ist die „GRIP-Runtime-Licence“ nötig. Zum Erstellen der Pro-
gramme ist noch zusätzlich die „GRIP-Development-License“ nötig. 
 
Es ist dringend zu empfehlen, während der Programmentwicklung die UNIGRAPHICS-
Online-Hilfe zur Verfügung zu haben, da hier die Syntax der einzelnen Kommandos nachge-
schlagen werden kann. 
 
GRIP-Programme waren in den letzten 20 Jahren sehr gut aufwärtskompatibel. Es laufen in 
UG NX2 noch Programme, die vor 20 Jahren mit UG V2 entwickelt wurden. Ggf. müssen dir 
Programme neu kompiliert und gelinkt werden. 
 
Die GRX-Dateien müssen im allgemeinen auf jeder Rechnerplatform (HP-UX, AIX, UNIX, 
Windows NT, ...) neu erzeugt werden. Es laufen jedoch auch Programme die z.B. auf einer 
HP-Maschine (HP-UX, UNIX) erzeugt wurden auch auf SGI (Silicon Graphics). 
 
Die 2. Programmiersprache zu UNIGRAPHICS nennt sich „User Function“ (UFUNC). Dabei 
handelt es sich um reine C-Programme. Als Compilor dient ein „normaler“ ANSI-fähiger 
Compilor (z.B. von Microsoft). Hier werden dem Programmierer zahlreiche Funktionen ange-
boten, welche UNIGRAPHICS-Funktionen darstellen. UFUNC geht weiter als GRIP, ist je-
doch aufwendiger zu programmieren. In UFUNC stehen dem Programmierer nahezu diesel-
ben Funktionen zur Verfügung, wie den Entwicklern von UNIGRAPHICS. Die Möglichkeiten 
von User Function sind umfangreicher wie die von GRIP. Beispielsweise ist es in GRIP nicht 
möglich, die „Part History“ eines Teiles zu löschen, in UFUNC gibt es eine Funktion dafür. 
Zahlreiche andere Beispiele könnten hier folgen. 
 
Der typische Weg beim Erstellen von GRIP-Programmen ist folgender: 
 
1) Programm mit einem ASCII-Editor schreiben  (Ergebnis: *.grs, GRIP Source) 
2) Programm übersetzen/kompilieren    (Ergebnis: *.gri,  GRIP Intermidiate) 
3) Programm binden/linken (nur das Hauptprog.) (Ergebnis: *.grx, GRIP Execute) 
 
Nur die „GRX-Datei“ ist für den Einsatz nötig (ggf. jedoch auch Partfiles oder Tabellen, aus 
denen Werte entnommen werden). 
 
 
Generelles Format eines GRIP-Statements: 
 
 Major word / parameters 
 
z.B. DELETE / ln1 
 
 
 Variable = Major word / parameters 
 
z.B.  p1 = POINT / 0,10,4.6 
 
  
Variable1 = Major word / Variable2, Minor word, Variable3 
 
z.B.  l1 = LINE / p1, LEFT, p2, TANTO, c1 

Deklarationen: 
 
 ENTITY / p1, L1, LINIE2, punkt(10)  grafische UG-Elemente, z.B. 
        Punkte, Linien, Flächen, Körper, 
        Bemaßungen, Text, ... 
 
 STRING / zeile(80), str(2), feld1(80,100) Textfelder 
 
 NUMBER / x1, ykoord(10), ss(10,10)  Werte, Zahlen,  Felder sind  
        zwingend anzugeben, Einzelwerte  
        optional 
 
 DATA / msg1, ‘Zeige Punkt’, $   Wertzuweisung für Werte und Texte, 
    msg2, ‘welche Farbe ?’   „$“ = Folgezeile 
 
 
Variablennamen: 
 
 • maximal 32 Zeichen 
 
 • erstes Zeichen muß ein Buchstabe sein 
 
 • kein Haupt- oder Nebenwort (Major- / Minor word) verwenden 
 
 • nur die ersten sechs Zeichen sind relevant (im Compilor-Listing) 
 
 • Klein- / Großschreibung ist egal 
 
 • Unterstrich „_“ ist erlaubt 
 
 
z.B: p1    ok 
 Linie_1   ok 
 LL1(4)    ok, einzelnes Feldelement 
 Koordinate_Startpunkt ok 
 
 1p    falsch, 1. Zeichen kein Buchstabe 
 LINE    falsch, GRIP-Sprachwort 
 Entity_Zaehler  falsch, 1. sechs Zeichen ein GRIP-Sprachwort 
 Dies_ist_auch_falsch_da_mehr_als_32_Zeichen 
 
 
Sonstige Syntax: 
 
Leerzeilen können beliebig viele verwendet werden. 
 
Kommentare beginne mit „$$“ (auch innerhalb einer Programmzeile möglich). 
 
Bei Folgezeilen ist das letzte Zeichen ein „$“. 
 
Innerhalb von Folgezeilen sind keine Kommentarzeilen erlaubt. 

Haupt- und Unterprogramme: 
 
 
Hauptprogramm 
 
 ENTITY/kreis 
 STRING/name1(30) 
 ... 
 $$--- Unterprogrammaufruf mit Uebergabeparameter 
 CALL/‘kreis_check‘,kreis,name1,retcod 
 
 IF retcode <> 0, JUMP/err007: 
 
 ... 
 HALT 
 
 
Unterprogramm „kreis_check.grs“ 
 
 PROC/kreis,name1,retcod 
 
 ENTITY/kreis  $$ Dimensionierung im Hauptprogramm 
 STRING/name1  $$ Dimensionierung im Hauptprogramm 
 
 ... 
 IF/LENF(name1) < 1, retcod = -1 
 ... 
 RETURN 
 
 
Die Variablen „kreis“ und „name1“ sind Eingabeparameter für das Unterprogramm 
„kreis_check“, die Variable „retcod“ ist ein Ausgabeparameter (Rückgabewert). 
 
Die Dimensionierung erfolgt im Hauptprogramm. 
 
Die Reihenfolge der Parameter muß im Unterprogrammaufruf und in der „PROC/-Zeile“ iden-
tisch sein, die Variablenname könnten voneinander abweichen. Nur die Reihenfolge zählt, 
nicht der Name! 
 
Alle Programme müssen kompiliert werden.  
 
Nur das Hauptprogramm wird gelinkt. 

Programmbeispiele für die interaktiven Kommandos: 
 
$$-+-+-+-+-+-+-+-+-+-+-+-+-+-+-+-+-+-+-+-+-+-+-+-+-+- 
$$ 
$$ GRIP - Kurs   / Walter Hogger /   8/96 
$$ 
$$-+-+-+-+-+-+-+-+-+-+-+-+-+-+-+-+-+-+-+-+-+-+-+-+-+- 
$$ 
$$ choose_demo.grs 
$$ 
$$ Beispiel des GRIP-Sprachwortes "CHOOSE". 
$$ 
$$ W. Hogger / 28.08.1996 
$$ 
 
start: 
 
 CHOOSE/'Waehle einen Wochentag !' ,  $ 
        'Montag',                $ 
        'Dienstag',              $  
        'Mittwoch',            $ 
        'Donnerstag',             $ 
        'Freitag',               $ 
        'Samstag',                $ 
        'Sonntag',                $ 
        '',                   $ 
        'ENDE',                 $ 
               DEFLT,1,antwort 
        JUMP/ende:,ende:,,,antwort 
 
  antwort = antwort - 4 
         
        IFTHEN/ antwort == 1 
          MESSG/'Die Woche ist noch lang ...' 
        ELSEIF/ antwort == 2 
          MESSG/'Lottospielen nicht vergessen !' 
        ELSEIF/ antwort == 3 
          MESSG/'Heute waer ein Feiertag recht ...' 
        ELSEIF/ antwort == 4 
          MESSG/'Nicht mehr lang ...' 
        ELSEIF/ antwort == 5 
          MESSG/'Endlich Freitag !' 
        ELSEIF/ antwort == 6 
          MESSG/'Was machste heute in der Fiiierma ?' 
        ELSEIF/ antwort == 7 
          MESSG/'Liest sowieso keiner am Sonntag !' 
        ELSE 
          MESSG/'Progammende' 
       JUMP/ende: 
        ENDIF  
 
$$---   nochmal ... 
    JUMP/start: 
 
ende:   HALT 

$$-+-+-+-+-+-+-+-+-+-+-+-+-+-+-+-+-+-+-+-+-+-+-+-+-+-+-+-+-+-+ 
$$ 
$$ GRIP - Kurs   / Walter Hogger /   8/96 
$$ 
$$-+-+-+-+-+-+-+-+-+-+-+-+-+-+-+-+-+-+-+-+-+-+-+-+-+-+-+-+-+-+ 
 
$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$ 
$$ 
$$ GRIP-Programm: param_demo.grs 
$$ 
$$------------------------------------------------------------ 
$$ 
$$ Das Programm dient als Beispiel fuer die Verwendung 
$$ des GRIP-Sprachwortes "PARAM/". 
$$ 
$$ Programmierer: W. Hogger 
$$ 
$$ Datum:         28.08.1996 
$$ 
$$ Aenderung:     - 
$$ 
$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$ 
$$ 
 
$$--- Deklarationen 
      DATA/ texth, 3.5  $$ Texthoehe 
      DATA/ schraege, 15  $$ Textschraegstellung 
$$$ DATA/ dehnung, 0.8     $$ Textdehnung (Kommentar!) 
 
start: 
 
$$---  Benutzerabfrage 
 PARAM/'Bitte die Text-Voreinstellung waehlen!', $ 
       'Texthoehe', texth,   $ 
       'Textneigung', INT, schraege,  $ 
       'Textdehnung', dehnung,   $ 
        ALTACT,'Hilfe-Menue', antwort 
 
$$$$$$$     back    cancel  ok   alternative  
 JUMP/ ende:,  ende: ,   ,  hilfe:,   antwort 
 
 
$$--- Einstellungen vornehmen 
        &CSIZE  = texth   $$ Texthoehe einstellen 
        &CSLANT = schraege $$ Textschraege einstellen 
        &ASPECT = dehnung $$ Textdehnung einstellen 
 
        JUMP/ende: 
 
 
$$---   Hilfemenue 
hilfe: 
 PRINT/'Hilfemenue des Programms: Texteinstellung' 
 PRINT/'___________________________________________' 
 PRINT/'' 
 PRINT/'... das ist doch wohl selbsterklaerend, odda !?' 
 
 JUMP/start: 
 
ende:   HALT 

$$-+-+-+-+-+-+-+-+-+-+-+-+-+-+-+-+-+-+-+-+-+-+-+-+-+- 
$$ 
$$ GRIP - Kurs   / Walter Hogger /   9/96 
$$ 
$$-+-+-+-+-+-+-+-+-+-+-+-+-+-+-+-+-+-+-+-+-+-+-+-+-+- 
$$ 
$$ mchoose_demo.grs 
$$ 
$$ Beispiel des GRIP-Sprachwortes "MCHOOS". 
$$ 
$$ W. Hogger / 25.09.1996 
$$ 
 
 STRING/Cuetext(40)  $$ Aufforderungstext 
 STRING/Angebot(7,40)    $$ Optionen 
 STRING/Alternat(40) $$ Alternate-Action-Text 
      NUMBER/Antwort(7)    $$ ARRAY fuer die  

$$ moeglichen Antworten 
 
$$---   Vorbelegungen 
 Cuetext    = 'Hier kann man knipsen ...' 
 Angebot(1) = 'Montag' 
 Angebot(2) = 'Dienstag' 
 Angebot(3) = 'Mittwoch' 
 Angebot(4) = 'Donnerstag' 
 Angebot(5) = 'Freitag' 
 Angebot(6) = 'Samstag' 
 Angebot(7) = 'Sonntag' 
 Alternat   = 'Alle Knoepfe wieder raus.' 
 
$$$      Antwort(1) = 1  $$ klappt nicht! d.h. die Knopfstellung 
   $$ kann nicht vorbelegt werden 
 
start: 
 MCHOOS/Cuetext,Angebot,Antwort,ALTACT,Alternat,antw1 
        JUMP/start:,ende:,,antw1 
 
 
$$--- Auswerten der Antworten 
$$---   0 = Knopf wurde nicht gedrueckt 
$$---   1 = Knopf wurde gedrueckt 
 DO/loop1:,i,1,7 
   IF/ Antwort(i) == 1,    $ 
   PRINT/'gewaehlt wurde -> ' + Angebot(i) 
loop1:  
 
 
$$---   nochmal ... 
    JUMP/start: 
 
ende:   HALT 

$$-+-+-+-+-+-+-+-+-+-+-+-+-+-+-+-+-+-+-+-+-+-+-+-+-+-+-+-+-+-+ 
$$ 
$$ GRIP - Kurs   / Walter Hogger /   8/96 
$$ 
$$-+-+-+-+-+-+-+-+-+-+-+-+-+-+-+-+-+-+-+-+-+-+-+-+-+-+-+-+-+-+ 
 
$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$ 
$$ 
$$ GRIP-Programm: crtwrt_demo.grs 
$$ 
$$------------------------------------------------------------ 
$$ 
$$ Das Programm dient als Beispiel fuer die Verwendung 
$$ der GRIP-Sprachworte "CRTWRT/" und "POS/". 
$$ 
$$ Programmierer: W. Hogger 
$$ 
$$ Datum:         29.08.1996 
$$ 
$$ Aenderung:     - 
$$ 
$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$ 
$$ 
 
$$--- Deklarationen 
 NUMBER/xyz(3)  $$ Feld fuer x-y-z-Koordinaten 
 STRING/meldung(40) $$ Text max. 40. Zeichen 
 
start: 
 
$$---   Benutzer soll Text eingeben 
 TEXT/'Welche Meldung willst du sehen ?',meldung,antw 
 JUMP/ ende:,  ende: ,      ,        ,           antw 
$$___________ back __ cancel __ ok __ alternative  
 
 
$$---   Falls nichts eingegeben wurde ... 
        IFTHEN/ LENF(meldung) < 1 
          MESSG/'Ok, wenn Du nicht willst ...',$ 
         'Programmabbruch' 
   JUMP/ende: 
 ENDIF 
 
 
$$---   Benutzer soll die Stelle der Meldung zeigen  
$$---  "POS" -> Bilschirmposition, "GPOS" -> Punktemenue 
 POS/'Wo soll die Meldung erscheinen ?', xyz,  antw 
 JUMP/ start:, ende: ,      ,        ,         antw 
$$___________ back __ cancel __ ok __ alternative  
 
 
$$---   Meldung anschreiben 
 CRTWRT/meldung,xyz 
 
 
$$---   Programm nocheinmal durchlaufen 
        JUMP/start: 
 
 
ende:   HALT 

Auszüge aus der Online – Hilfe: Major Words 
 
 
 

Major Word Function Description Category 

ABSF Absolute Value System/Calculator 

ACOSF Angle Whose Cosine Is arg System/Calculator 

ADDPL Add Row System/Assemblies 

ADIM Angular Dimensioning/Drafting Objects 

ANGLF Angle of a Line System/Verify Properties 

ANLSIS Rotate about XC or YC Axis System/3D Analysis 

ANLSIS Project along ZC Axis System/3D Analysis 

ANLSIS Bounded by Sheets System/3D Analysis 

ANLSIS Thin Shell System/3D Analysis 

ANLSIS Mass Properties of Solid Bodies System/3D Analysis 

ANLSIS Arc Length System/3D Analysis 

ANLSIS 2D Analysis System/Verify Properties 

APPEND File Pointer Control File/Edit 

ARCDIM Arc Length Dimensioning/Drafting Objects 

ASATT Assign Object Attributes System/Attributes 

ASCENT Associated Object of Dim/Draft Object System/3D Analysis 

ASCII ASCII Value of a Character Strings 

ASGNFT Assign Font System/Parameters 

ASINF Angle Whose Sine Is arg System/Calculator 

ATANF Angle Whose Tangent Is arg System/Calculator 

AUTOSF Create Sheets from Curves Sheet Body/Body Surface 

Auszüge aus der Online – Hilfe: Minor Words 
 
 
Minor Word List 
 
ACCRCY  
ACTIVE  
ADD  
AFILL  
AFTER  
AITPZ  
ALL  
ALTACT  
ALWAYS  
AMBI  
ANCHOR  
ANGLE  
ANTALI  
APEX  
APPEND  
APPROX  
ARCLEN  
ARROW  
ARROWS  
ASCEND  
ATANGL  
ATRBTS  
AUTO  
AXIS  
BACK  
BACKZ  
BEFORE  
BLANK  
BLEND  
BMDRY  
BND  
BOTH  

Auszüge aus der Online – Hilfe: GPA Symbols 
 
 
 

GPA Symbol Function Description Category 

&ABS Absolute Coordinate System Geometry 

&ACTPRT Active Part Status File Management 

&ALARND All Around Symbol Display Drafting 

&ANGTOL Modeling Angle Tolerance Geometry 

&APSITE Appended Text Site Dimension 

&ARCCTR Radius Dimension Display Dimension 

&ARDSP Arrow Display Dimension 

&ARFLDT Filled Arrowhead Distance Dimension 

&ARIANG Arrowhead Included Angle Dimension 

&ARLNSM Arc Length Symbol Dimension 

&ARROW  Arrowhead 1 Type Dimension 

&ARROW2 Arrowhead 2 Type Dimension 

&ASIZE Arrowhead Size Dimension 

&ASPECT Aspect Ratio Drafting 

&ATTDIS Attribute Name Display Attributes 

&ATTYPE Attribute Type Attributes 

&AUNT Angular Dimension Units Dimension 

&BATCH Current Execution Mode System 

&BGCLR Background Color System 

&CAREA Customer Area of the File Header File Management 

&CFONT Character Font Dimension 

&CHAIN Chaining Method Chain 

&CHTOL Chaining Tolerance Chain 

Auszüge aus der Online – Hilfe: EDA Symbols 
 
 
 

 EDA Symbol Function Description Category 

&ALLARN All Around Symbol Display Drafting & Dimensioning 

&ALNSYM Arc Length Symbol Display Drafting & Dimensioning 

&ARWDAT Arrowhead Display Data Drafting & Dimensioning 

&ASCCNT Number of Associated Objects Drafting & Dimensioning 

&ATDISL Name Display Location (X,Y,Z) Attributes 

&ATTALL All Object Attributes Attributes 

&ATTTL Attribute Title Attributes 

&ATTVL Attribute Value Attributes 

&BLANK Blank Geometry/Spline 

&CCOUNT Number of Component Members 
Geometry/Components & Reference  

Sets 

&CDATE Component Date 
Geometry/Components & Reference  
Sets 

&CHRDAT Character Parameters Drafting & Dimensioning 

&CHRSET Native Language Menu Character Sets Geometry/Spline 

&CLRNME Color Name Geometry/Spline 

&CLRNUM Color Number Geometry/Spline 

&CMEM 
Object Identifier of Nth Member of  

Component 

Geometry/Components & Reference  

Sets 

&COLOR Color Geometry/Spline 

&CPNAM Component Part Name 
Geometry/Components & Reference 
 Sets 

&CRSNAM Reference Set Name 
Geometry/Components & Reference  

Sets 

&CSCALE Component Scale 
Geometry/Components & Reference  
Sets 

 
 


